


ATARI® 400/800™

PILOT DEMONSTRATION PROGRAMS

USERS GUIDE



A Warner Communications Company



Use with
ATARI® 400™ or ATARI® 800™
PERSONAL COMPUTER SYSTEMS

PILOT DEMONSTRATION PROGRAMS USERS GUIDE



A Warner Communications Company

Every effort has been made to ensure that this manual accurately documents this product of the ATARI Computer Division. However, because of the ongoing improvement and update of the computer software and hardware, ATARI, INC. cannot guarantee the accuracy of printed material after the date of publication and cannot accept responsibility for errors or omissions.

Reproduction is forbidden without the specific written permission of ATARI, INC. Sunnyvale, CA 94086. No right to reproduce this document, nor the subject matter thereof, is granted unless by written agreement with, or written permission from the Corporation.

PRINTED IN U.S.A.

PROGRAM CONTENTS © 1980 ATARI, INC.
MANUAL © 1981 ATARI, INC.

CONTENTS

INTRODUCTION	v
1 LOADING THE PROGRAMS	1
2 PILOT PROGRAMS FOR CHILDREN	3
3 A PILOT TEACHING PROGRAM	5
4 PILOT "TURTLE GRAPHICS" DEMONSTRATION	7
5 PILOT DO-IT-YOURSELF SLIDE SHOW	9

INTRODUCTION

Two cassette tapes are included with your **ATARI® PILOT** cartridge. They contain demonstration programs that illustrate different educational and recreational applications for PILOT. In many ways, PILOT is an easier language for computer programming than BASIC. In fact, PILOT is so easy to learn, that some of the enclosed demonstration programs were written by elementary school children guided by their teachers. Aside from small changes to adapt them to your ATARI Computer's capabilities, these programs remain as written.

Each of these programs show different facets of PILOT. We chose them because they are relatively simple to understand, and we hope they will help you write your own PILOT programs. After all, the goal of ATARI PILOT is to put you in control of your computer.

LOADING THE PROGRAMS




Figure 1 Loading Diagram

1. Connect the **ATARI 400™** or **ATARI 800™** Personal Computer System to your television set and to a wall outlet, as instructed in the operator's manual.
2. Make sure that the **ATARI 410™** Program Recorder is properly connected to the computer console and to a wall outlet. See your *ATARI 410 Program Recorder Operator's Manual* for further details if necessary.

Note: If you have "daisy-chained" peripherals to your computer console (connecting one unit to another) and do not wish to disconnect them, connect your ATARI 410 Program Recorder to the I/O CONNECTOR of the last unit in the chain.

If an ATARI Disk Drive is used, the Disk Operating System (DOS) and system software use some of available RAM. The amount of RAM used varies with the particular version of DOS that you are using. This overhead needs to be taken into account when calculating the amount of RAM required to run a program.

3. Make sure that at least 16K of RAM is installed in your ATARI Computer. (See the *ATARI 800 Operator's Manual* for **ATARI Memory Module™** loading instructions.)

-
4. Open the cartridge door on the top of the computer console and insert the ATARI PILOT cartridge into the cartridge slot (the LEFT CARTRIDGE slot in the ATARI 800 Computer). Close the cartridge door.
 5. Turn on your television set.
 6. Turn on your computer console by pressing the POWER switch on the right side of the console to ON.
 7. If all equipment is properly connected and turned on, your television screen should be displaying the READY prompt with the white square "cursor" just below.
 8. Press **STOP/EJECT** on your ATARI 410 Program Recorder to open the cassette door.
 9. Hold the demonstration program cassette so that the desired side is up and the tape leader is facing you.
 10. Slide the cassette into the cassette holder and close the door.
 11. If necessary, press **REWIND** and rewind the tape to the beginning. When the tape is rewound, press **STOP/EJECT**.
 12. Type **LOAD C:** on the computer keyboard and press **RETURN**. The computer will "beep" once to remind you to press **PLAY** on the program recorder.

Note: For those of you already familiar with BASIC, PILOT does **not** use a quotation mark (") before the C: in the LOAD command.

13. Press **PLAY** on the program recorder and then the **RETURN** key on the computer console to start the loading process. Through the window of the program recorder, note that the tape is turning. The "beeps" and other sounds you hear coming from the television speaker tell you that the program is being loaded into computer memory.

Note: If you have problems loading the program, and if you have other peripherals in addition to the program recorder attached to the computer console, try disconnecting the other peripherals and connecting the program recorder directly into the computer console to isolate the problem.

14. When the television screen displays the READY prompt or the cursor again, type **RUN** and press **RETURN**.

PILOT PROGRAMS FOR CHILDREN

Cassette A, Side 1

USING THE PROGRAMS

After you load the program from the tape and type **RUN**, you will see a small graphics display on your television screen followed by a Menu with two choices. If you type **1** and press **RETURN**, you will be given the chance to help create your own version of **GOLDILOCKS**. When the computer stops in the middle of typing a sentence, you should fill in the rest of the sentence with any response you want. After you press **RETURN**, the story will continue, sometimes incorporating your responses in later parts of the story. It is fun to see how the story changes as different people use this program. Once the story is finished, you will automatically be returned to the Main Menu.

If you type **2** and press **RETURN**, you will be using a program that lets you associate colors, textures, and sounds with various objects. You should let your mind run free when selecting your answers. You might be surprised at the result!

BACKGROUND

The two programs on this side of Cassette 1 were written by participants of a special educational project at Stanford Research Institute (now SRI International) in 1969. This was probably the first project to use the **PILOT** language with teachers and with students in the early elementary grades. It successfully demonstrated the ease with which children can learn to write programs in this language. These programs have since been used in Montessori schools, public schools, computer access centers (such as the University of California's Lawrence Hall of Science), as well as for teacher training.

The first program, **GOLDILOCKS**, is a special version of **GOLDILOCKS AND THE THREE BEARS** adapted for **PILOT** by Dr. Dean Brown. The program allows the user to insert his or her own character names, feelings, and plot. The computer provides the general framework for the story. The computer weaves the user's responses back into the story — sometimes with unexpected results! Unlike many computer-based educational programs, this program never makes value judgments. It assumes the user is in control of the program and is perfectly capable of making the best possible choice of responses.

Because of the importance of stories to children's intellectual development, we recommend using **GOLDILOCKS** as an example for writing your own stories. The structure of the program is very simple. Basically, it uses only three commands:

- | | |
|---------------------|---|
| T: | The TYPE command places text on the display screen. |
| A:\$RESPONSE | The ACCEPT command accepts input from the keyboard and places it in the variable, \$RESPONSE . |
| PA:#P | The PAUSE command stops the continuation of the program for a period of time, making the text easier to read. The #P can be replaced by a number instead of a numeric variable. |

You may wish to write similar types of stories based on other childhood favorites such as JACK AND THE BEANSTALK or LITTLE RED RIDING HOOD. Use your imagination and humor; but most of all, let your children guide you.

The second program, COLORS AND FEELINGS, was written by children and teachers during the SRI Education Laboratory Summer School Project in 1969. The purpose of this program is to encourage children to use their imagination in making associations between colors and objects, people, places, and qualities. As with GOLDBLOCKS, there are no "right" or "wrong" answers in this program. If you are interested in knowing more about how "metaphoric" thinking can be used in education, you may wish to refer to some of the literature on creativity. Some of the materials found in books on synectics offer great potential for adaptation into PILOT programs.

In addition to the commands T and A, this program uses "pointer" variables to change the background color of the screen. ATARI PILOT has the facility to change the color of the screen to any of 128 colors. Each color is made from one of 16 hues and one of 8 luminosities (brightness). To change the screen color, use the compute (C:) command to assign a color value to memory location 710 (the location of the background color information). The command C:@710=?\256 will generate a random color change. If you want to experiment with a given color value, you should replace the "?\256" with any even number from 0 to 254.

A PILOT TEACHING PROGRAM

Cassette A, Side 2

USING THE PROGRAM

After you load the program from the tape and type **RUN**, you will see a small graphics display followed by a request for your name. As with the previous two programs, you should respond to the computer's request for information by typing your name and pressing **RETURN**. Unlike the previous two programs, you will find that this program pays attention to your responses and replies accordingly!

BACKGROUND

This program also came out of the work done at SRI in 1969. It makes use of the **M:** (**MATCH**) command to check for anticipated responses, but still avoids making judgments on the "correctness" of the responses.

To understand this program, you should look carefully at a listing. To see the program press the **SYSTEM RESE** key, type **LIST**, and press **RETURN**. To make the list pause before running off the top of the screen, press the **CTRL** key and the **1** key at the same time. Press these keys again to continue the listing.

Pay special attention to the **M:** commands. Each of these commands contains a list of anticipated responses which the program will be looking for. Notice that you don't always have to spell out each and every possible response in **PILOT**. Rather, by being clever in checking these responses, you may only need to use part of a word or a phrase. For example, looking for the string **MIL** (i.e., **M: MIL**), would let **PILOT** find a match with **MILK**, **MILLION**, **MILES**, etc. However, within the context of this particular dialogue, only **MILK** is likely to occur.

Note that spaces are a very important part of the match command. If you put a space in front of the match item, you will only match words that start with the item. For example, look at the following two match commands:

M: MIL, WAT, CREAM, JUIC

M: AIR, WIN, GAS, PIL

In the first example, a match will only be found if the referenced strings are at the beginning of a word, since each item is preceded by a space. In the second example, the referenced strings will be matched if they occur anywhere in the response.

The match command is probably the major distinguishing feature between **PILOT** and other programming languages. It is a key to writing computer-aided instruction programs because you usually want to know if a response is reasonable or not, given the context of the question.



PILOT "TURTLE GRAPHICS" DEMONSTRATION

Cassette B, Side 1

USING THE PROGRAM

After you load the program from the tape and type **RUN**, you will see a small graphics display followed by the display of a Menu with two choices. If you type **1** and press **RETURN**, you will enter the CITY GROW program that draws houses on the screen in random locations and colors. As each house is being drawn, you can see the PILOT graphics commands that are being executed at that moment. By pressing the **START** key, you can speed up the drawing of houses; by pressing **START** again, the drawing speed is slowed back down.

Press the **SELECT** key to return to the Main Menu. Menu choice number 2 switches you to the PRETTY LINES program, giving you access to two more graphics demonstrations. One demonstration generates a graphical ballet with colored lines, while the other draws various geometric patterns based on "cycloids." These programs also use sound, so be sure your television speaker is turned up!

Press the **SELECT** key at any time to stop running any of these graphics demonstrations and return to the Main Menu.

BACKGROUND

The original PILOT language was designed for computers that did not have the graphics capabilities of your ATARI 400 or 800 Personal Computer System. Therefore, ATARI has made some very powerful graphics commands available in PILOT so that you can draw pictures and designs, present visual information, and learn some simple and powerful ways to describe geometric shapes. These unique graphics capabilities are based on a system called "Turtle Graphics," developed originally by Dr. Seymour Papert and the LOGO Project at the Massachusetts Institute of Technology.

In Turtle Graphics, all points and lines drawn on the screen are a result of an imaginary robot, called a "turtle," which leaves traces of its movements on the television display. This robot turtle only obeys a limited number of commands, such as:

GOTO X,Y	Go to the point located at coordinates X,Y.
TURNT0 N	Turn the turtle so it is pointing N degrees clockwise from the point corresponding to 12:00 on a clock.
GO N	Go N units in your current direction.
TURN N	Turn by N degrees clockwise from the current direction.
DRAW N	Draw a line, N units long, in the current direction.

These commands are described in detail in the *PILOT Primer: An Instruction Manual for the PILOT Programming Language* and *The Student PILOT: A Reference Guide for Students*.



PILOT DO-IT-YOURSELF SLIDE SHOW

Cassette B, Side 2

USING THE PROGRAM

After you load the program from the tape and type **RUN**, you will see a small graphics display followed by the first of a series of slides describing PILOT. You may advance the slides by pressing any console key, by pressing the button on a joystick controller, or by just waiting.

As you can see from the slides, ATARI PILOT lets you gain access to several of the different ATARI Computer text modes. By using these larger characters, you can make very interesting slides for your next presentation!

BACKGROUND

Business people, teachers, and many others often give presentations using overhead or slide projectors. Much of the content of these presentations involves text organized into simple charts or tables. This program allows you to use the ATARI Computer and a television set to make these presentations. Because of the different text modes and sound capabilities available in PILOT, this is a very flexible medium for this purpose.

To best understand this program, look at the listing shown below. Lines 10-1390 contain the program itself; all lines after 1400 contain the material for the slides. You may easily modify the contents of a slide by locating its label (*SLIDE1, etc.) and then modifying the text to be displayed. Each slide is composed of a certain number of lines of text designated by \$L1, \$L2, etc. (e.g., see lines 1590-1640). These string variables (\$L1, \$L2, etc.) correspond to lines of text on the screen.

There are two main modes used by this program — mode 1 and mode 2.

Mode 1 (*INIT1) is composed of 24 lines, each capable of displaying up to 20 characters of text. Thus, if you wanted to place the message "HI THERE!" on the top line, you would use a statement like this:

```
1590 C:#L1 = HI THERE!
```

Blank lines are inserted if there are no contents for a particular line number.

Mode 2 (*INIT2) is similar to mode 1, except that you may use up to 12 lines of 20 characters each. Each slide is composed of the following sequence:

```
*SLIDE1      slide number label
U:*INIT1     or *INIT2 -- display mode
C:$L1 =
.
.
.
U:*DISP      displays the slide
U:*WAIT      waits for a button or key press
E:           ends the slide
```

Notice that the sequence of commands, U:*SLIDE1, U:*SLIDE2, and so on in lines 1300-1400 of the program determine the order in which the slides will be presented. Thus, if you wanted to repeat a previous slide, you would insert another U: (USE) command for that slide in the desired place in the slide sequence. Similarly, to delete a slide, just delete the line on which the slide is "used." To delete *SLIDE5 from the show, delete the line containing U:*SLIDE5.

It is suggested that you SAVE the first part of this program separately from the rest if you are going to make your own slide shows. To do that, type

SAVE C: 10,1390 if you are using the Program Recorder
SAVE D:SLIDE 10,1390 if you are using a disk drive

You can load these programs, add the rest of your slide show to them, and then save the finished show for use later. If you want to save the slide descriptions, then type:

SAVE C: 1400,9999
or
SAVE D:SHOW 1400,9999

By loading each portion of the program separately without typing NEW in between, you can reassemble the complete program. This is because ATARI PILOT does not erase an existing program when loading a new one. This feature should be used with caution, making certain that each separate portion contains no overlap in line numbers with the other portions.

To add graphics to a slide, write a module that contains the graphics you want (see the PILOT manuals for details on this) and place the appropriate U: (USE) command in the proper place so that the graphics slide will appear in the correct sequence.

For more information on graphics, please refer to either *The PILOT Primer: An Instruction Manual for the PILOT Programming Language* or *The Student PILOT: A Reference Guide for Students*.

SLIDE SHOW LISTING

```
10 GR: CLEAR
20 T:
30 POS: 10, 2
40 T: WELCOME TO ATARI PILOT
50 :
60 GR: DRAWTO 0, 8; TURNTO 17
70 C: #N=0
80 *L
90 C: #N=#N+1
100 GR: DRAW #N; TURN 91
110 SO: #N
120 C: #C=(#N\3)+1
130 J(#C=2): *Y
140 J(#C=3): *B
150 GR: PEN RED
160 J: *D
170 *Y
180 GR: PEN YELLOW
190 J: *D
```

```

200 *B
210 GR: PEN BLUE
220 *D
230 J(#N<74):*L
240 GR: DRAW #N-3
250 T:
260 POS: 14,2
270 T: PRESS RETURN\
280 A:
290 *RET
300 GR: QUIT
310 R:
320 R: ATARI PILOT SLIDE SHOW
330 R:
340 R: VARIABLES FOR COLOR REGISTER ADDRESSES
350 R:  #A      BACKGROUND
360 R:  #B      UPPERCASE, NUMBERS
370 R:  #C      INVERSE UPPERCASE, NUMBERS
380 R:  #D      LOWERCASE
390 R:  #E      INVERSE LOWERCASE
400 R: OTHER VARIABLES
410 R:  #F      DEBOUNCING 'BUTTON'
420 R:  #G      GRAPHICS MODE FLAG (1 or 2)
430 R:  #H,#I   TEMPORARY VALUES
440 R:  #P      TIME TO PAUSE
450 R:          1 = APPROX. 1/10 SECOND
460 R:  $L,$L1, . . . , $L24  SCREEN LINES
470 R:  @B77  RESET DELAY FOR FLASHING SCREEN
480 R: SUBROUTINES
490 R:  *INIT1  SET GRAPHICS MODE 1
500 R:          INIT $L1, . . . , $L24
510 R:  *INIT2  SET GRAPHICS MODE 2
520 R:          INIT $L1, . . . , $L12
530 R:  *CLR    CLEAR DISPLAY SCREEN
540 R:  *DISP   DISPLAY SCREEN LINES IN MODE #G
550 R:  *WAIT   WAIT FOR KEYSTROKE OR 'BUTTON'
560 R:  *PAUSE  PAUSE FOR #P TIME OR
570 R:          KEYSTROKE OR 'BUTTON'
580 R:  *TEST   TEST FOR KEYSTROKE OR 'BUTTON'
590 R:
600 C: #A=712
610 C: #B=708
620 C: #C=710
630 C: #D=711
640 C: #F=0
650 J: *START
660 *INIT1 C: #G=1      [*INIT1
670 J: *INIT
680 *INIT2 C: #G=2      [*INIT2
690 *INIT  C: #H=(3-#G)*12
700 *INITLP
710 C: $L=L#H
720 C: $$L=
730 C: #H=#H-1
740 J(#H):*INITLP
750 E:
760 *CLR          [*CLR
770 WRITE: S
780 E:
790 *DISP         [*DISP
800 C(#G<>1):#G=2
810 C: #H=(3-#G)*12
820 C: #I=1
830 C: @B1373=#Q
840 C: @B1374=#G
850 *DISPL
860 C: $L=L#I
870 WRITE: S, $$L

```

```

880 C:#I=#I+1
890 J(#I<=#H):*DISPL
900 CLOSE:5
910 E:
920 *WAIT [*WAIT
930 U:*TEST
940 JN:*WAIT
950 E:
960 *PAUSE [*PAUSE
970 C:#H=#P
980 *PAUSELP
990 U:*TEST
1000 EY:
1010 C:#H=#H-1
1020 J(#H):*PAUSELP
1030 E:
1040 *TEST [*TEST
1050 C:@B77=0 [RESET FLASHING SCREEN DELAY
1060 R:
1070 R:@B19 IS CLOCK LOCATION
1080 J(@B19>(2*(3-#G))):*YES [CHECK CLOCK
1090 A:= [INIT ACCEPT BUFFER
1100 J(@B764<255):*YES [KEYSTROKE
1110 J(@B53279=5):*RET
1120 J(#F=%T8):*NOI 'BUTTON' HAS NOT CHANGED
1130 C:#F=%T8 [SET NEW 'BUTTON' STATE
1140 J(#F):*YES ['BUTTON' JUST ON
1150 *NO M:N [MATCH FAILS
1160 E:
1170 *YES M:, [MATCH SUCCEEDS
1180 C:@B764=255 [CLEAR KEYBOARD
1190 C:@19=0 [ZERO REAL TIME CLOCK CTR
1200 S0:20
1210 PA:4
1220 S0:
1230 E:
1240 R:
1250 R:
1260 *START [*START
1270 C:#Q=16 [SET TEXT WINDOW
1280 C:@19=0 [ZERO REAL TIME CLOCK CTR
1290 R:*****SLIDE TABLE*****
1300 U:*SLIDE1
1310 C:#Q=0 [NO TEXT WINDOW
1320 U:*SLIDE2
1330 U:*SLIDE3
1340 U:*SLIDE4
1350 U:*SLIDE5
1360 U:*SLIDE6
1370 U:*SLIDE7
1380 U:*SLIDE8
1390 J:*RET
1400 R:
1410 R:*****
1420 R: SLIDES BEGIN HERE
1430 R:*****
1440 R:***USE INIT1 FOR 24 ROWS*****
1450 R:***USE INIT2 FOR 12 ROWS*****
1460 R:
1470 R:***COLOR CORRESPONDENCE TABLE**
1480 R: 'FONT' COLOR
1490 R:
1500 R: UPPERCASE YELLOW
1510 R: lowercase GREEN
1520 R: INVERSE CAPS BLUE
1530 R: inverse lowr RED
1540 R:
1550 R:*****

```



```

1560 *SLIDE1
1570 R:
1580 U:*INIT2
1590 C:$L2= INTRODUCING
1600 C:$L4= Programmed
1610 C:$L5= Inquiry
1620 C:$L6= Learning
1630 C:$L7= Or
1640 C:$L8= Teaching
1650 U:*DISP
1660 T: PRESS ANY KEY FOR MORE SLIDES
1670 T: OR PRESS JOYSTICK'S FIRE BUTTON
1680 T: ( TIMED AUTO-ADVANCE IN EFFECT )
1690 U:*WAIT
1700 E:
1710 *SLIDE2
1720 U:*INIT2
1730 C:$L2= A LANGUAGE FOR:
1740 C:$L4= * children
1750 C:$L6= * teachers
1760 C:$L8= * parents
1770 U:*DISP
1780 U:*WAIT
1790 E:
1800 *SLIDE3
1810 U:*INIT2
1820 C:$L2= EXPERIENCE
1830 U:*DISP
1840 C:#P=15
1850 U:*PAUSE
1860 C:$L3= + EXPLORATION
1870 U:*DISP
1880 U:*PAUSE
1890 C:$L4= + DISCOVERY
1900 U:*DISP
1910 U:*PAUSE
1920 C:*$L5= + HUMOR
1930 U:*DISP
1940 U:*PAUSE
1950 C:$L7= =====
1960 U:*DISP
1970 U:*PAUSE
1980 C:$L9= computer
1990 C:$L10= assisted
2000 C:$L11= imagination
2010 U:*DISP
2020 C:@B#A=192
2030 C:@B#B=132
2040 C:@B#C=54
2050 C:@B#D=10
2060 U:*WAIT
2070 E:
2080 *SLIDE4
2090 U:*INIT2
2100 C:$L1= advantages of
2110 C:$L2= pilot
2120 C:$L4= easy to learn
2130 C:$L6= TRANSPORTABLE
2140 C:$L8= conversational
2150 C:$L10= PEOPLE-ORIENTED
2160 U:*DISP
2170 U:*WAIT
2180 E:
2190 *SLIDES
2200 U:*INIT1
2210 C:$L2= core pilot
2220 C:#L3= commands
2230 C:#L6= T: TYPE A MESSAGE

```

```

2240 C:$L8= A:ACCEPT AN ANSWER
2250 C:$L10= M:FIND A MATCH
2260 C:$L12= C:COMPUTE A RESULT
2270 C:$L14= J:JUMP TO *LABEL
2280 C:$L16= U:USE A PROCEDURE
2290 C:$L18= E:END A PROCEDURE
2300 C:$L20= R:REMARK ONLY
2310 U:*DISP
2320 U:*WAIT
2330 E:
2340 *SLIDE6
2350 U:*INIT1
2360 C:$L2= pilot variables
2370 C:$L5= # NUMERIC
2380 C:$L7= $ STRING
2390 C:$L10= pilot conditioners
2400 C:$L13= Y TRUE
2410 C:$L15= N FALSE
2420 C:$L18= logical operators
2430 C:$L20= (<=> (<=>)= (<=>))
2440 U:*DISP
2450 U:*WAIT
2460 E:
2470 *SLIDE7
2480 U:*INIT1
2490 C:$L2= graphics
2500 C:$L5= relative absolute
2510 C:$L7= DRAW N DRAWTO X,Y
2520 C:$L8= TURN N TURNT0 N
2530 C:$L9= GO N GOTO X,Y
2540 C:$L10= FILL N FILLTO X,Y
2550 C:$L13= CLEAR
2560 C:$L14= QUIT
2570 C:$L16= PEN UP,ERASE, red
2580 C:$L17= YELLOW,BLUE
2590 U:*DISP
2600 U:*WAIT
2610 E:
2620 *SLIDE8
2630 U*INIT1
2640 C:$L2= sound: 4 voices
2650 C:$L5= S0: U1 U2 U3 U4
2660 C:$L7= EACH VOICE IS
2670 C:$L9= REPRESENTED BY A
2680 C:$L11= NUMBER FROM 1-31
2690 C:$L15= THE TONAL RANGE
2700 C:$L17= IS 2 1/2 OCTAVES
2710 C:$L19= IN HALF-STEPS
2720 U:*DISP
2730 S0:1
2740 PA:30
2750 S0:1,5
2760 PA:30
2770 S0:1,5,8
2780 PA:30
2790 S0:1,5,8,12
2800 PA:120
2810 S0:0
2820 U:*WAIT
2830 E:

```

**LIMITED 90-DAY WARRANTY
ON ATARI® PERSONAL COMPUTER PRODUCTS**

ATARI, INC. ("ATARI") warrants to the original consumer purchaser that this ATARI Personal Computer Product (not including computer programs) shall be free from any defects in material or workmanship for a period of 90 days from the date of purchase. If any such defect is discovered within the warranty period, ATARI's sole obligation will be to repair or replace, at its election, the Computer Product free of charge on receipt of the unit (charges prepaid, if mailed or shipped) with proof of date of purchase satisfactory to ATARI at any authorized ATARI Service Center. For the location of an authorized ATARI Service Center nearest you, call toll-free:

In California (800) 672-1430
Continental U.S. (800) 538-8547

or write to: Atari, Inc.
Customer Service Department
1340 Bordeaux Drive
Sunnyvale, CA 94086

YOU MUST RETURN DEFECTIVE COMPUTER PRODUCTS TO AN AUTHORIZED ATARI SERVICE CENTER FOR IN-WARRANTY REPAIR.

This warranty shall not apply if the Computer Product: (i) has been misused or shows signs of excessive wear, (ii) has been damaged by being used with any products not supplied by ATARI, or (iii) has been damaged by being serviced or modified by anyone other than an authorized ATARI Service Center.

ANY APPLICABLE IMPLIED WARRANTIES, INCLUDING WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE, ARE HEREBY LIMITED TO NINETY DAYS FROM THE DATE OF PURCHASE. CONSEQUENTIAL OR INCIDENTAL DAMAGES RESULTING FROM A BREACH OF ANY APPLICABLE EXPRESS OR IMPLIED WARRANTIES ARE HEREBY EXCLUDED. Some states do not allow limitations on how long an implied warranty lasts or do not allow the exclusion or limitation of incidental or consequential damages, so the above limitations or exclusions may not apply to you.

This warranty gives you specific legal rights and you may also have other rights which vary from state to state.

**DISCLAIMER OF WARRANTY
ON ATARI COMPUTER PROGRAMS**

All ATARI computer programs are distributed on an "as is" basis without warranty of any kind. The entire risk as to the quality and performance of such programs is with the purchaser. Should the programs prove defective following their purchase, the purchaser and not the manufacturer, distributor, or retailer assumes the entire cost of all necessary servicing or repair.

ATARI shall have no liability or responsibility to a purchaser, customer, or any other person or entity with respect to any liability, loss, or damage caused directly or indirectly by computer programs sold by ATARI. This disclaimer includes but is not limited to any interruption of service, loss of business or anticipatory profits or consequential damages resulting from the use or operation of such computer programs.

REPAIR SERVICE

If your ATARI Personal Computer Product requires repair other than under warranty, please contact your local authorized ATARI Service Center for repair information.

IMPORTANT: If you ship your ATARI Personal Computer Product, package it securely and ship it, charges prepaid and insured, by parcel post or United Parcel Service.



PRINTED IN U.S.A.



A Warner Communications Company

C017810 REV.1